

MÉTODOS DOS QUADRADOS MÍNIMOS GENERALIZADOS PARA SISTEMATIZAÇÃO DE TERRAS PARA IRRIGAÇÃO POR SUPERFÍCIE: ALGORITMO EM BASIC

DOURADO NETO, D. (*)

MAIA, P. C. S. (**)

CAIXETA, A. V. (***)

RESUMO

O Algoritmo em BASIC proposto determina a equação do plano que melhor se adapta às condições naturais do terreno, determina as matrizes de cotas calculadas e alturas de corte e de aterro, antes e depois do rebaixamento, e calcula o rebaixamento com a respectiva relação corte/aterro adotada. O método dos quadrados mínimos generalizado permite a minimização de revolvimento de volume de terra por unidade de área, apresentando a vantagem de ter sido desenvolvido analiticamente, ser programável e prever condições práticas de área irregular.

Palavras-chave: Sistematização. Irrigação por superfície. Método dos quadrados mínimos generalizado.

Generalized least-square method for

land grading for surface irrigation: Algorithm in BASIC

SUMMARY

GENERALIZED LEAST-SQUARE METHOD FOR LAND GRADING FOR SURFACE IRRIGATION: ALGORITHM IN BASIC

A computer program for land grading design procedures for solving the plane-of-best-fit equation in an irregular field is presented in BASIC with determination of grid point elevation matrix, computation of cuts and fills before and after the adjustment increment and the cut-fill ratio. The least-square method is used to minimize the earthwork quantities for a specified cut-fill ratio per area unit.

Key-words: Land grading. Surface irrigation. Least-square method.

1. INTRODUÇÃO

O sistema de irrigação superficial se caracteriza pela condução da água pela superfície do solo, apresentando as vantagens de possuir um elevado potencial para redução de consumo de energia, não interferir nos tratamentos fitossanitários, independe de quaisquer parâmetros climáticos e da altura das plantas, além de apresentar simplicidade operacional. Possui as limitações de

ter acentuada dependência de condições topográficas, ser inadequado para solos excessivamente permeáveis, requerer maior nível de conhecimento (informações técnicas para operar o sistema), o qual é intransferível, possuir divulgação limitada pela indústria e pelos técnicos e demandar mais mão-de-obra que os outros sistemas.

Apesar da irrigação por superfície ocupar cerca de 80% dos duzentos e vinte milhões de hectares irrigados no mundo, e quase esse mesmo percentual nos quase três milhões de hectares irrigados no Brasil, esse sistema vem apresentando, especialmente nos projetos públicos brasileiros níveis relativamente baixos de eficiência de aplicação e armazenagem devido a vários fatores, desde problemas de sistematização até o dimensionamento do projeto.

Após serem executados os estudos preliminares de Pedologia e a área possuir aptidão agrícola e ser constatada a viabilidade da irrigação por superfície, far-se-á sistematização do terreno, que possui o objetivo de conseguir a uniformidade topográfica com o gradiente de declive que melhor se adapta às condições naturais do terreno, trazendo alguns benefícios como a melhor drenagem superficial, maior uniformidade na distribuição de água, fertilizante e semente, menor utilização de mão-de-obra e minimização dos efeitos deletérios da erosão, devendo algumas providências iniciais serem

(*) Professor Doutor. Departamento de Agricultura. ESALQ. USP C.P. 9. Piracicaba - SP.

(**) Engenheiro Agrônomo. Primeira Superintendência Regional. Companhia de Desenvolvimento do Vale do São Francisco. Montes Claros - MG.

(***) Engenheiro Civil. Aluno de Pós-Graduação. Departamento de Engenharia Rural. C.P. 9. ESALQ. USP. Piracicaba - SP.

tomadas, tais como o reconhecimento da área, escolha do período das operações, eliminação de obstáculos superficiais, pré-sistematização, levantamento plani-altimétrico da área, divisão em sub-áreas e escolha do método de cálculo.

O presente trabalho tem por objetivos apresentar um algoritmo em BASIC para viabilizar o uso do método dos quadrados mínimos generalizado (recursos computacionais) e transmitir alguns aspectos básicos fundamentais para sistematização de terras para irrigação por superfície.

2. MATERIAIS E MÉTODOS

Algumas providências iniciais devem ser tomadas, tais como: reconhecimento da área, escolha da época das operações, eliminação de obstáculos superficiais, pré-sistematização, levantamento plani-altimétrico da área, divisão da área em sub-áreas.

O reconhecimento da área envolve a observação da topografia, drenagem superficial e aspectos relacionados à sistematização como profundidade do solo agricultável, presença de camadas compactadas, natureza da variação textural e níveis de salinidade e alcalinidade na sequência dos horizontes.

A execução dos serviços de sistematização não deve ser feita na época chuvosa devido menor rendimento dos trabalhos, maior compactação e maior custo com transporte de água. A época recomendável é o período mais seco do ano. A amplitude do período indica o número de máquinas ou o tamanho da área a trabalhar.

Alguns obstáculos, tais como raízes, tocos, pedras e vegetações quaisquer, devem ser eliminados no intuito de melhorar a qualidade do serviço.

A área deve ser gradeada e pré-sistematizada no intuito de evitar a presença de irregularidades superficiais

que provoque o assentamento de estacas, no levantamento plani-altimétrico, em pontos não representativos da área considerada comprometendo os cálculos.

O estaqueamento da área é feito obedecendo alguns critérios como a fixação do espaçamento entre estacas em função da regularidade topográfica da área, da precisão desejada, do custo operacional, e a determinação das linhas básicas ou eixos imaginários (x e y) ortogonais.

O espaçamento mais frequentemente utilizado é o de 20 x 20 m, resultando numa área representativa por estaca de 400 m² e um total de 25 estacas.ha⁻¹. Espaçamentos menores aumentam o custo operacional, a precisão e o tempo de execução do serviço.

A estaca padronizada possui as seguintes dimensões: 1,20 m x 0,05 m x 0,01 m.

Convencionou-se que as estacas que representam áreas de corte devem ser pintadas de vermelho na extremidade superior, do ápice para base, com o comprimento equivalente a altura de corte, e as estacas que representam áreas de aterro devem ser pintadas de azul na base, de baixo para cima, com o comprimento equivalente à altura de aterro. Estacas que não representam nem áreas de corte, nem áreas de aterro, não são pintadas. Ainda, convencionou-se que a anotação dos dados deve ser feita da seguinte forma:

COTA ORIGINAL	LEITURA DE MIRA
COTA CALCULADA	CORTE (-) OU ATERRO (+)

Caso a área a ser sistematizada seja demasiadamente grande, a sua divisão em sub-áreas é imprescindível. Alguns critérios devem ser

adotados como a tendência das curvas de nível quanto a direção e a distância entre si.

3. RESULTADOS E DISCUSSÃO

O custo de sistematização é diretamente proporcional ao volume de terra revolvido. O volume de terra revolvido por unidade de área diminui com a diminuição da área a ser sistematizada.

Os volumes máximos de corte e de aterro variam para cada solo, dependendo da relação corte - aterro (R) e da altura máxima admissível de corte e de aterro (Tabela 1).

$$R = \frac{V_C}{V_A} \quad (1)$$

onde: R refere-se à relação corte - aterro ($1,1 \leq R \leq 2,0$); V_C ao volume de corte (m³); e V_A ao volume de aterro (m³).

Assumindo que a altura média máxima de corte é igual à de aterro, e reescrevendo a equação (1), tem-se o seguinte sistema de equações:

$$V_C + V_A = 10000 \cdot K \cdot (m^3 \cdot ha^{-1}) \quad (2)$$

$$V_C - R \cdot V_A = 0 \quad (3)$$

onde K refere-se à altura (m) média máxima admissível de corte, assumindo-a equivalente à de aterro.

Aplicando a regra de Kramer:

$$D = \begin{vmatrix} 1 & 1 \\ 1 & -R \end{vmatrix} = -(1+R) \quad (4)$$

$$D_1 = \begin{vmatrix} 10000 \cdot K & 1 \\ 0 & -R \end{vmatrix} = -10000 \cdot K \cdot R \quad (5)$$

$$D_2 = \begin{vmatrix} 1 & 10000 \cdot K \\ 1 & 0 \end{vmatrix} = -10000 \cdot K \quad (6)$$

$$V_C = \frac{D_1}{D} \quad (7)$$

$$V_A = \frac{D_2}{D} \quad (8)$$

Sendo assim, conclui-se que:

$$V_C = \frac{10000 \cdot K \cdot R}{1+R} \quad (9)$$

$$V_A = \frac{10000}{1+R} \quad (10)$$

Tabela 1 - Estimativas dos volumes máximos de corte (V_c , $m^3.ha^{-1}$) e de aterro (V_a , $m^3.ha^{-1}$) para diferentes relações corte - aterro (R) e para duas alturas (K) médias admissíveis de corte.

R	K = 0,10 m		K = 0,15 m	
	V_c ($m^3.ha^{-1}$)	V_a ($m^3.ha^{-1}$)	V_c ($m^3.ha^{-1}$)	V_a ($m^3.ha^{-1}$)
1,1	523,81	476,19	785,71	714,29
1,2	545,45	454,55	818,18	681,82
1,3	565,22	434,78	847,83	652,17
1,4	583,33	416,67	875,00	625,00
1,5	600,00	400,00	900,00	600,00
1,6	615,38	384,62	923,08	576,92
1,7	629,63	370,37	944,00	555,56
1,8	642,89	357,14	964,29	535,71
1,9	655,17	344,83	982,76	517,24
2,0	666,67	333,33	1000,00	500,00

Tem-se que a equação do plano que melhor se adapta às condições naturais do terreno é dada pela seguinte expressão:

$$H(x, y) = A + Bx + Cy \quad (11)$$

onde $H(x, y)$ refere-se à cota calculada da estaca na posição de abscissa x e ordenada y (m); A à cota calculada da estaca na abscissa "0" e ordenada "0" (m); B ao coeficiente angular da reta que representa o gradiente de declive que melhor se adapta às condições naturais do terreno na direção x (m/espacamento na direção y), (espacamento em metros); C ao coeficiente angular da reta que representa o gradiente de declive que melhor se adapta às condições naturais do terreno na direção y (m/espacamento na direção x), (espacamento em metros).

Sendo o resíduo calculado por:

$$r(x, y) = h(x, y) - H(x, y) \quad (12)$$

onde: $r(x, y)$ = resíduo ou altura de corte (+) ou altura de aterro (-) (m); $h(x, y)$ = cota original do terreno obtida no levantamento topo-

gráfico plani-altimétrico (m).

Com a determinação da equação do plano que melhor se adapta às condições naturais do terreno, faz com que a soma dos resíduos seja nula.

$$\sum_{y=1}^m \sum_{x=1}^n r(x, y) = 0 \quad (13)$$

Porém, a soma dos quadrados do resíduo (RQ) não é.

$$RQ = \sum_{y=1}^m \sum_{x=1}^n [h(x, y) - H(x, y)]^2 \quad (14)$$

substituindo $H(x, y)$ (Equação 11):

$$RQ = \sum_{y=1}^m \sum_{x=1}^n [h(x, y) - A - Bx - Cy]^2 \quad (15)$$

Para condição de mínimo, tem-se que a derivada primeira deve ser nula e a segunda positiva:

$$\frac{\partial RQ}{\partial A} = \frac{\partial RQ}{\partial B} = \frac{\partial RQ}{\partial C} = 0$$

Sendo assim, tem-se que:

$$\frac{\partial RQ}{\partial A} = -2 \sum_{y=1}^m \sum_{x=1}^n [h(x, y) - A - Bx - Cy] = 0 \quad (16)$$

$$\frac{\partial RQ}{\partial B} = -2 \sum_{y=1}^m \sum_{x=1}^n x [h(x, y) - A - Bx - Cy] = 0 \quad (17)$$

$$\frac{\partial RQ}{\partial C} = -2 \sum_{y=1}^m \sum_{x=1}^n y [h(x, y) - A - Bx - Cy] = 0 \quad (18)$$

Originando o seguinte sistema de equações:

$$\sum_{y=1}^m \sum_{x=1}^n A + B \sum_{y=1}^m \sum_{x=1}^n x + C \sum_{y=1}^m \sum_{x=1}^n y = \sum_{y=1}^m \sum_{x=1}^n h(x, y) \quad (19)$$

$$A \sum_{y=1}^m \sum_{x=1}^n x + B \sum_{y=1}^m \sum_{x=1}^n x^2 + C \sum_{y=1}^m \sum_{x=1}^n xy = \sum_{y=1}^m \sum_{x=1}^n xh(x, y) \quad (20)$$

$$A \sum_{y=1}^m \sum_{x=1}^n y + B \sum_{y=1}^m \sum_{x=1}^n xy + C \sum_{y=1}^m \sum_{x=1}^n y^2 = \sum_{y=1}^m \sum_{x=1}^n yh(x, y) \quad (21)$$

Sendo:

$$\sum_{y=1}^m \sum_{x=1}^n A = N \cdot A \quad (N = m \cdot n) \quad (22)$$

Aplicando a Regra de Cramer, tem-se que:

$$C_1 = \left(\sum_{y=1}^m \sum_{x=1}^n x^2 \right) \cdot \left(\sum_{y=1}^m \sum_{x=1}^n y^2 \right) - \left(\sum_{y=1}^m \sum_{x=1}^n xy \right)^2 \quad (23)$$

$$C_2 = \left(\sum_{j=1}^n \sum_{k=1}^n y \right) \left(\sum_{j=1}^n \sum_{k=1}^n xy \right) - \left(\sum_{j=1}^n \sum_{k=1}^n x \right) \left(\sum_{j=1}^n \sum_{k=1}^n y^2 \right) \quad (24)$$

$$C_3 = \left(\sum_{j=1}^n \sum_{k=1}^n x \right) \left(\sum_{j=1}^n \sum_{k=1}^n xy \right) - \left(\sum_{j=1}^n \sum_{k=1}^n y \right) \left(\sum_{j=1}^n \sum_{k=1}^n x^2 \right) \quad (25)$$

$$C_4 = \left(\sum_{j=1}^n \sum_{k=1}^n x \right) \left(\sum_{j=1}^n \sum_{k=1}^n y \right) - N \left(\sum_{j=1}^n \sum_{k=1}^n xy \right) \quad (26)$$

$$D = NC_1 + 2 \left[\left(\sum_{j=1}^n \sum_{k=1}^n x \right) \left(\sum_{j=1}^n \sum_{k=1}^n y \right) \left(\sum_{j=1}^n \sum_{k=1}^n xy \right) - \left(\sum_{j=1}^n \sum_{k=1}^n x^2 \right) - \left(\sum_{j=1}^n \sum_{k=1}^n y^2 \right) - \left(\sum_{j=1}^n \sum_{k=1}^n x^2 \right) \right] \quad (27)$$

$$D_1 = C_1 \left[\sum_{j=1}^n \sum_{k=1}^n h(x, y) \right] + C_2 \left[\sum_{j=1}^n \sum_{k=1}^n xh(x, y) \right] + C_3 \left[\sum_{j=1}^n \sum_{k=1}^n yh(x, y) \right] \quad (28)$$

$$D_2 = C_2 \left[\sum_{j=1}^n \sum_{k=1}^n h(x, y) \right] + C_4 \left[\sum_{j=1}^n \sum_{k=1}^n xh(x, y) \right] + \left[\sum_{j=1}^n \sum_{k=1}^n xh(x, y) \right] \left[n \left(\sum_{j=1}^n \sum_{k=1}^n y^2 \right) - \left(\sum_{j=1}^n \sum_{k=1}^n y \right)^2 \right] \quad (29)$$

$$D_3 = C_3 \left[\sum_{j=1}^n \sum_{k=1}^n h(x, y) \right] + C_4 \left[\sum_{j=1}^n \sum_{k=1}^n xh(x, y) \right] +$$

$$\left[\sum_{j=1}^n \sum_{k=1}^n yh(x, y) \right] \left[n \left(\sum_{j=1}^n \sum_{k=1}^n x^2 \right) - \left(\sum_{j=1}^n \sum_{k=1}^n x \right)^2 \right] \quad (30)$$

$$A = \frac{D_1}{D} \quad (31)$$

$$B = \frac{D_2}{D} \quad (32)$$

$$C = \frac{D_3}{D} \quad (33)$$

Verificando a derivada segunda, tem-se que:

$$\frac{\partial^2 RQ}{\partial A^2} \Big|_{A,C} = 2 \quad (34)$$

$$\frac{\partial^2 RQ}{\partial B^2} \Big|_{A,C} = 2 \sum_{j=1}^n \sum_{k=1}^n x^2 \quad (35)$$

$$\frac{\partial^2 RQ}{\partial C^2} \Big|_{A,B} = 2 \sum_{j=1}^n \sum_{k=1}^n y^2 \quad (36)$$

O ajuste das alturas de corte e de aterro se torna necessário devido à compactação nas áreas de corte e de aterro com a passagem das máqui-

nas e o excesso de terra para diques (no caso de irrigação por faixas).

A relação corte - aterro (R)(Equação 1) é dada pela razão entre o volume de corte e de aterro que varia de 1,1 a 2,0, oscilando na grande maioria dos nossos solos na faixa de 1,2 a 1,5. Quanto maior for o teor de areia e maior for a altura média de corte, menor será o valor de R, e quanto maior for o teor de argila ou de matéria orgânica e menor for a altura de corte, maior será o valor de R.

Devido às causas anteriormente mencionadas, o rebaixamento (RB) do plano de sistematização se torna necessário:

$$R = \frac{V_c}{V_a} = \frac{A_c}{A_a} = \frac{\left(\sum_{j=1}^g C_j \cdot RB \cdot N_c \right)}{\left(\sum_{j=1}^h a_j \cdot RB \cdot N_a \right)} \quad (37)$$

onde: A_c refere-se à área de corte (m); A_a à área de aterro (m); C_i à altura de corte (m); a_i à altura de aterro (m); N_c ao número de estacas com corte; N_a ao número de estacas com aterro; RB ao rebaixamento do plano de sistematização (m).

Assumindo, por simplicidade de cálculo, que:

$$\frac{A_c}{A_a} = 1 \quad (38)$$

Tem-se que:

$$RB = \frac{R \sum_{j=1}^h a_j - \sum_{j=1}^g C_j}{R N_a + N_c} \quad (39)$$

A estimativa dos volumes de corte e de aterro, que não é objetivo deste trabalho é de fundamental importância para contratação dos serviços de sistematização.

Existem vários métodos para estimativa do volume de corte e de aterro, dentre eles tem-se o método das áreas médias, fórmula prismoidal, método dos quatro pontos e método do somatório.

Os dois primeiros requerem a utilização de recursos computacionais e são bastante precisos, sendo o método das áreas médias o mais recomendável. O método dos quatro pontos apesar de sua concepção e utilização mais simples, na prática tem apresentado resultados não muito diferentes. Porém, o método do somatório só é recomendável para estimativas rápidas e grosseiras dos volumes de corte e aterro, pois normalmente os superestimam.

O presente algoritmo foi feito em BASIC, onde os dados de entrada são os seguintes: 1. Número de colunas (I) da matriz interna; 2. Número de linhas (J) da matriz interna; 3. Cotas originais do campo $[H(x,y)]$ (m). As leituras devem ser feitas por colunas, de cima para baixo e da esquerda para a direita, excluindo as marginais (caso tenham sido lidas no campo); 4. Espaçamento na direção x (EX) (m); 5. Espaçamento na direção y (EY) (m); 6. Cotas originais totais no campo, incluindo as marginais $[HT(x,y)]$ (m) (caso suas respectivas leituras tenham sido feitas).

Toda matriz irregular deve ser transformada em matriz "regular" e nas estacas de cotas inexistentes será lido "0" (zero). Na impressão da matriz, nos pontos ausentes, será impresso (*) asterisco.

O valor numérico das cotas originais existentes nunca pode dar nulo. Nas estacas onde se darão cortes ou aterros, serão impressos nas respectivas matrizes, um número negativo (-) para corte (c), ou positivo (+) para aterro (a). No caso de não revolvimento de terra será impresso "0" (zero).

Caso tenham sido lidas as cotas marginais, serão impressas as suas respectivas cotas calculadas e alturas de corte e de aterro, antes e depois do rebaixamento, nas respectivas matrizes, sendo criadas e duas linhas e duas colunas marginais. A impressão dar-se-á como uma matriz "regular", pois não são

previstas as diferentes distâncias marginais. Também devem ser lidos valores nulos para cotas inexistentes.

A equação que melhor se adapta às condições naturais do terreno não leva em consideração as cotas

marginais, mesmo que suas leituras tenham sido feitas. Na verdade, tanto o espaçamento regular, os espaçamentos marginais e as cotas marginais efetivamente terão utilidade para o cálculo do volume de corte e de aterro.

O rebaixamento (RB) foi calculado conforme a equação 39.

A relação corte-aterro (R) deve ser lida conforme conveniência do projeto, observando as características peculiares de cada solo.

Tabela 2 - Exemplo elucidativo de uma área irregular, com leituras de cotas marginais e malha de 20x30m, resultando, aproximadamente, 17 estacas.ha⁻¹.

[I]									
3,733									
		[1]	[2]	[3]	[4]	[5]	[6]	[7]	
(-)	(2)	(6,8)	(6,2)	(6,9)	(6,9)	(-)	(-)	(-)	(3,8)
	11m	23m	24m	19 m	18 m				
[1]	(6,7)	6,6	6,1	6,7	6,7	-	-	-	(-)
	15m		20m						
[2]	(5,8)	5,7	5,4	5,9	6,0	-	-	-	(-)
	17m		30m						
[3]	(4,6)	4,7	4,4	5,0	5,4	-	-	-	(-)
	17m							17m	
[4]	(3,9)	4,0	3,8	4,2	4,4	4,2	3,9	3,8	(3,8)
[J]4,867	13m			3,404 (centróide)				16m	
[5]	(3,5)	3,4	3,1	3,5	3,4	3,6	3,2	2,9	(2,9)
	10m							15m	
[6]	(2,9)	2,7	2,2	3,2	2,6	2,4	2,3	2,2	(2,1)
								15m	
[7]	(-)	-	1,5	2,0	1,6	1,8	1,7	1,5	(1,6)
								16m	
[8]	(-)	-	0,7	1,2	0,7	1,0	1,0	0,9	(1,0)
			22m	21m	16m	15m	13m	5m	
(-)		(-)	(0,8)	(1,4)	(0,9)	(1,2)	(1,1)	(0,8)	(-)

1 matriz interna

2 cotas marginais

Nesse exemplo específico, obteve-se a seguinte equação do plano que melhor se adapta às condições naturais do terreno:

$$HC(x,y) = 7,325 - 1,047.10^{-3}.x - 0,798.y(40)$$

Sendo assim, tem-se o rebaixamento (RB):

$$RB = 0,031 \text{ m}$$

Portanto, a equação do plano depois do rebaixamento é dada

pela seguinte expressão:

$$HR(x,y) = 7,325 - 1,047.10^{-3}.x - 0,798.y - 0,031(41)$$

Tabela 3 - Matriz de cotas calculadas sem rebaixamento (com marginais) (plano que melhor se adapta às condições naturais de terreno)

*	7,31	7,30	7,29	7,28	*	*	*	*
6,53	6,52	6,51	6,50	6,49	*	*	*	*
5,73	5,72	5,71	5,70	5,69	*	*	*	*
4,93	4,92	4,91	4,90	4,89	*	*	*	*
4,13	4,12	4,11	4,10	4,09	4,08	4,07	4,06	4,06
3,34	3,33	3,32	3,31	3,30	3,28	3,27	3,26	3,26
2,54	2,53	2,52	2,51	2,50	2,49	2,48	2,47	2,47
*	*	1,72	1,71	1,70	1,69	1,68	1,67	1,67
*	*	0,92	0,91	0,90	0,89	0,88	0,87	0,87
*	*	0,92	0,91	0,90	0,89	0,88	0,87	*

Tabela 4 - Matriz de alturas de corte e de aterro sem rebaixamento (com marginais)

*	0,51	1,10	0,39	0,38	*	*	*	*
-0,17	-0,08	0,41	-0,20	-0,21	*	*	*	*
-0,07	0,02	0,31	-0,20	-0,31	*	*	*	*
0,33	0,22	0,51	-0,10	-0,51	*	*	*	*
0,23	0,12	0,31	-0,10	-0,31	-0,12	0,17	0,26	0,26
-0,16	-0,07	0,22	-0,19	-0,10	-0,32	0,07	0,36	0,36
-0,36	-0,17	0,32	-0,69	-0,10	0,09	0,18	0,27	0,37
*	*	0,22	-0,29	0,10	-0,11	-0,02	0,17	0,07
*	*	0,22	-0,29	0,20	-0,11	-0,12	-0,03	-0,13
*	*	0,12	-0,49	0,00	-0,31	-0,22	0,07	*

4. CONCLUSÕES

Em função do que foi abordado, verifica-se que: i. o método dos quadrados mínimos generalizado permite obter gradientes de declive que mais se adequam às condições naturais do terreno, resultando num

menor revolvimento de volume de terra por unidade de área e, conseqüentemente, menor custo e menor depalperamento físico do solo sistematizado; ii. a utilização de recursos computacionais é de fundamental importância para otimização de tempo e uso de métodos mais consistentes e precisos.

REFERÊNCIAS BIBLIOGRÁFICAS

- BERNARDO, S. Manual de irrigação. 2. ed. Viçosa, Imprensa Universitária da UFV, 1982.
- SCALOPPI, E. J. Sistema de Irrigação por superfície. Informe Agropecuário, Belo Horizonte, 12(139): julho de 1986.

6. Adendo

```

10 '*****'
20 'DURVAL DOURADO NETO/ANICÉSIO VIEIRA CAIXETA/PEDRO CÉSAR MAIA
30 'PROGRAMA: QMG.BAS
40 'MÉTODO DOS QUADRADOS MÍNIMOS GENERALISADO
50 'CÁLCULOS: I- EQUAÇÃO DO PLANO QUE MELHOR SE ADAPTA
60 '      II- MATRIZ DE COTAS CALCULADAS SEM REBAIXAMENTO
70 '      III- MATRIZ DE ALTURA DE CORTE E ALTURAS DE ATERRO
80 '          ANTES DO REBAIXAMENTO DO PLANO
90 '      IV- REBAIXAMENTO
100 '      V- EQUAÇÃO DO PLANO DEPOIS DO REBAIXAMENTO
110 '      VI- MATRIZ DE ALTURAS DE CORTE E ALTURAS DE
120 '          ATERRO DEPOIS DO REBAIXAMENTO DO PLANO
130 '      VII- MATRIZ DE COTAS CALCULADAS DEPOIS DO
140 '          REBAIXAMENTO DO PLANO
150 '*****'
160 'EQUAÇÃO DO PLANO
170 PRINT "MATRIZ INTERNA-LEITURA SEM AS COTAS MARGINAIS, CASO TENHAM SIDO LIDAS
NO LEVANTAMENTO PLANI-ALTIMÉTRICO"
180 PRINT "NÚMERO DE COLUNAS=";
190 INPUT I
200 PRINT "NÚMERO DE LINHAS=";
210 INPUT J
220 INPUT "ESPAÇAMENTO NA DIREÇÃO X (EX) (EM METROS)=" ,EX
230 INPUT "ESPAÇAMENTO NA DIREÇÃO Y (EY) (EM METROS)=" ,EY
240 DIM H(I+2,J+2)
250 PRINT "LEIA (CTRL) PARA COTAS NÃO EXISTENTES EM MATRIZ
INCOMPLETA"
260 FOR X=1 TO I
270 FOR Y=1 TO J
280 PRINT "ENTRE COM H(";X;",";Y;")=";
290 INPUT H(X,Y)
300 IF H(X,Y) <> 0 THEN GOTO 330
310 D=D+1
320 GOTO 380
330 N=I*J-D
340 E=X*Y+E
350 F=H(X,Y)+F
360 G=X*H(X,Y)+G
370 K=Y*H(X,Y)+K
380 NEXT Y
390 NEXT X
400 FOR X=1 TO I
410 FOR Y=1 TO J
420 IF H(X,Y)=0 THEN L=L+1
430 NEXT Y
440 Z = Y-L-1
450 P = Z*X+P
460 Q = Z*X^2+Q
470 L = 0
480 NEXT X
490 FOR Y = 1 TO J
500 FOR X = 1 TO I

```

```

510 IF H(X,Y)=0 THEN M = M+1
520 NEXT X
530 W = X-M-1
540 U = W*Y+U
550 V = W*Y^2+V
560 M = 0
570 NEXT Y
580 DT = N*Q*V+2*U*P*E-Q*U^2-V*P^2-N*E^2
590 DT1 = P*E*K+Q*V*F+U*G*E-F*E^2-P*G*V-Q*U*K
600 DT2 = N*G*V+F*E*U+U*P*K-G*U^2-K*E*N-V*P*F
610 DT3 = -U*Q*F-E*G*N-K*P^2+N*Q*K+P*G*U+F*P*E
620 A = DT1/DT
630 B = DT2/DT
640 C = DT3/DT
650 PRINT "ENTRE COM (1) PARA MATRIZ SEM COTAS MARGINAIS OU (2) PARA MATRIZ COM
COTAS MARGINAIS=";
660 INPUT MM
670 LPRINT; TAB(1); "I- EQUAÇÃO DO PLANO QUE MELHOR SE ADAPTA"
680 LPRINT:LPRINT; TAB(1); "HC(X,Y) = (";A;") + (";B;") * X + (";C;") * Y"
690 LPRINT:LPRINT
700 '*****
710 'CORTE/ATERRO SEM REBAIXAMENTO COM (OU SEM) COTAS MARGINAIS
720 DIM R(I+2,J+2), HC(I+2,J+2), VX(I+2), VY(J+2), HT(I+2,J+2), RT(I+2,J+2), HCT(I+2,J+2)
730 IF MM = 1 THEN GOTO 810
740 FOR X = 2 TO I+1
750 VX(X) = X-1
760 NEXT X
770 FOR Y = 2 TO J + 1
780 VY(Y) = Y - 1
790 NEXT Y
800 GOTO 870
810 FOR X = 1 TO I
820 VX(X) = X
830 NEXT X
840 FOR Y = 1 TO J
850 VY(Y) = Y
860 NEXT Y
870 IF MM = 1 THEN GOTO 950
880 PRINT "MATRIZ TOTAL"
890 FOR X = 1 TO I + 2
900 FOR Y = 1 TO J + 2
910 PRINT "ENTRE COM HT(";X;",";Y;") = ";
920 INPUT HT(X,Y)
930 NEXT Y,X
940 GOTO 1010
950 FOR X = 1 TO I
960 FOR Y = 1 TO J
970 HT(X,Y) = H(X,Y)
980 NEXT Y
990 NEXT X
1000 IF MM = 1 THEN I=I-2:J=J-2
1010 FOR X = 1 TO I+2
1020 FOR Y = 1 TO J+2
1030 IF MM = 1 THEN GOTO 1210

```

```

1040 IF X = 1 AND Y = 1 THEN GOTO 1130
1050 IF X = 1 AND Y = J+2 THEN GOTO 1140
1060 IF X = I+2 AND Y = 1 THEN GOTO 1150
1070 IF X = I+2 AND Y = J+2 THEN GOTO 1160
1080 IF X = 1 THEN GOTO 1170
1090 IF X = I+2 THEN GOTO 1180
1100 IF Y = 1 THEN GOTO 1190
1110 IF Y = J+2 THEN GOTO 1200
1120 GOTO 1210
1130 VX(1) = PP(X,Y) / EX:VY(1) = TT(X,Y) / EY:GOTO 1210
1140 VX(1) = PP(X,Y) / EX:VY(J+2) = TT(X,Y) / EY+J:GOTO 1210
1150 VX(I+2) = PP(X,Y) / EX+I:VY(1) = TT(X,Y) / EY:GOTO 1210
1160 VX(I+2) = PP(X,Y) / EX+I:VY(J+2) = TT(X,Y) / EY+J:GOTO 1210
1170 VX(1) = PP(X,Y) / EX:GOTO 1210
1180 VX(I+2) = PP(X,J) / EX+I:GOTO 1210
1190 VY(1) = TT(X,Y) / EY:GOTO 1210
1200 VY(J+2) = TT(X,Y) / EY+J
1210 IF HT(X,Y) = 0 THEN HCT(X,Y) = 0:GOTO 1230
1220 HCT(X,Y) = A + B * VX(X) + C * VY(Y)
1230 RT(X,Y) = HCT(X,Y) - HT(X,Y)
1240 NEXT Y,X
1250 LPRINT
1260 IF MM = 1 THEN LPRINT; TAB(1); "II)MATRIZ DE COTAS CALCULADAS SEM REBAIXAMENTO
(SEM MARGINAIS)" : GOTO 1280
1270 LPRINT; TAB(1); "II) MATRIZ DE COTAS CALCULADAS SEM REBAIXAMENTO (COM MARGINAIS)"
1280 LPRINT;TAB(1);"(PLANO QUE MELHOR SE ADAPTA AS CONDIÇÕES NATURAIS DO
TERRENO)"
1290 LPRINT:LPRINT
1300 FOR Y=1 TO Y+2
1310 W = 1
1320 FOR X = 1 TO I + 2
1330 IF HCT(X,Y) = 0 THEN LPRINT TAB(W);" * ";;GOTO 1350
1340 LPRINT TAB(W);USING "###.##";HCT(X,Y);
1350 W = W + 10
1360 NEXT X,Y
1370 IF MM = 1 THEN GOTO 1480
1380 LPRINT:LPRINT
1390 LPRINT:LPRINT:LPRINT;TAB(1);"III) MATRIZ DE ALTURAS DE CORTE E DE ATERRO
SEM REBAIXAMENTO (COM MARGINAIS)"
1400 LPRINT:LPRINT
1410 FOR Y = 1 TO J+2
1420 W = 1
1430 FOR X = 1 TO I+2
1440 IF HT(X,Y) = 0 THEN LPRINT TAB(W);" * ";;GOTO 1460
1450 LPRINT TAB(W);USING "###.##";RT(X,Y);
1460 W = W+10
1470 NEXT X,Y
1480 '*****
1490 'REBAIXAMENTO DELTA
1500 REM CÁLCULO CORTE E ATERRO
1510 ST = 11
1520 LPRINT:LPRINT:LPRINT:LPRINT
1530 IF MM = 1 THEN LPRINT;TAB(1); "III)MATRIZ DE ALTURAS DE CORTE E DE ATERRO
SEM REBAIXAMENTO (SEM MARGINAIS)" :GOTO 1550

```

```

1540 LPRINT;TAB(1);"III-A) MATRIZ DE ALTURAS DE CORTE E DE ATERRO SEM
REBAIXAMENTO (SEM MARGINAIS)"
1550 LPRINT:LPRINT
1560 IF MM = 1 THEN I=I+2:J=J+2
1570 FOR Y = 1 TO J
1580 FOR X = 1 TO I
1590 HC(X,Y)= A+B*X+C*Y
1600 R(X,Y)= HC(C,Y)-H(X,Y)
1610 R(X,Y)= INT((R(X,Y))*100)/100
1620 W = (X*ST+1)-ST
1630 LPRINT TAB(W+ST-LEN$(STR$(R(X,Y))));
1640 IF H(X,Y) = 0 THEN LPRINT;" * ";:GOTO 1680
1650 IF R(X,Y) < 0 THEN LPRINT;R(X,Y);"C";
1660 IF R(X,Y) = 0 THEN LPRINT;R(X,Y);
1670 IF R(X,Y)>0 THEN LPRINT;R(X,Y);"A";
1680 NEXT X
1690 NEXT Y
1700 LPRINT:LPRINT
1710 PRINT "ENTRE COM R=";
1720 INPUT R
1730 FOR X = 1 TO I
1740 FOR Y = 1 TO J
1750 IF H(X,Y) = 0 THEN GOTO 1840
1760 IF R(X,Y) = 0 THEN GOTO 1840
1770 IF R(X,Y) > 0 THEN GOTO 1790
1780 IF R(X,Y) < 0 THEN GOTO 1820
1790 NA = NA + 1
1800 SA = SA + R(X,Y)
1810 GOTO 1840
1820 NC = NC + 1
1830 SC = SC - R(X,Y)
1840 NEXT Y
1850 NEXT X
1860 DELTA = (R*SA-SC)/(R*NA+NC)
1870 LPRINT:LPRINT:LPRINT;TAB(1);"IV- REBAIXAMENTO DELTA"
1880 LPRINT:LPRINT;TAB(1);"DELTA=";DELTA:LPRINT:LPRINT:LPRINT
1890 LPRINT;TAB(1);"V- EQUAÇÃO DO PLANO DEPOIS DO REBAIXAMENTO"
1900 LPRINT:LPRINT; TAB(1); "HR(X,Y) = (";A;") + (";B;")*X + (";C;") * Y - (";ABS(DELTA);)"
1910 LPRINT:LPRINT:LPRINT
1920 '*****
1930 'CORTE/ATERRO COM REBAIXAMENTO
1940 DIM HR(I+2,J+2),RC(I+2,Y+2)
1950 IF MM=1 THEN LPRINT;TAB(1);"VI) MATRIZ DE ALTURAS DEFINITIVAS DE CORTE E DE
ATERRO COM REBAIXAMENTO (SEM MARGINAIS)":GOTO 1970
1960 LPRINT; TAB(1);"VI) MATRIZ DE ALTURAS DEFINITIVAS DE CORTE E DE ATERRO COM
REBAIXAMENTO (COM MARGINAIS)"
1970 LPRINT:LPRINT
1980 IF MM = 1 THEN I=I-2:J=J-2
1990 FOR Y = 1 TO J+2
2000 FOR X = 1 TO I+2
2010 IF HCT(X,Y) = 0 THEN HR(X,Y)=0:GOTO 2030
2020 HR(X,Y) = HCT(X,Y)-DELTA
2030 IF RT(X,Y) = 0 THEN RCT(X,Y)=0:GOTO 2060
2040 RCT(X,Y) = RT(X,Y)-DELTA

```


```

2050 RCT(X,Y) = INT((RCT(X,Y)*100)/100
2060 W = (X*ST+1)-ST
2070 LPRINT TAB(W+ST-LEN$(STR$(RC(X,Y))));
2080 IF RCT(X,Y) = HR(X,Y) THEN LPRINT;" * ";;GOTO 2130
2090 IF RCT(X,Y) < 0 THEN LPRINT;RCT(X,Y);"C";
2100 IF RT(X,Y) = 0 THEN LPRINT;" * ";;GOTO 2120
2110 IF RCT(X,Y) = 0 THEN LPRINT;RCT(X,Y);
2120 IF RCT(X,Y)>0 THEN LPRINT;RCT(X,Y);"A";
2130 NEXT X
2140 NEXT Y
2150 LPRINT:LPRINT
2160 LPRINT:LPRINT
2170 IF MM = 1 THEN LPRINT;TAB(1);"VII) MATRIZ DE COTAS CALCULADAS DEFINITIVAS COM
REBAIXAMENTO (SEM MARGINAIS)":GOTO 2190
2180 LPRINT; TAB(1);"VII)MATRIZ DE COTAS CALCULADAS DEFINITIVAS COM REBAIXAMENTO (COM
MARGINAIS)"
2190 LPRINT:LPRINT
2200 FOR Y = 1 TO J+2
2210 FOR X = 1 TO I+2
2220 HR(X,Y)=INT(HR(X,Y)*100)/100
2230 Z = (X*ST+1)-ST
2240 LPRINT TAB(Z+ST-LEN$(STR$(HR(X,Y))));
2250 IF HR(X,Y) = 0 THEN LPRINT;" * ";;GOTO 2270
2260 LPRINT; HR(X,Y);
2270 NEXT X
2280 NEXT Y
2290 LPRINT:LPRINT:LPRINT:LPRINT;TAB(1);"ESPAÇAMENTO NA DIREÇÃO X= ";EX;"(M)"
2300 LPRINT;TAB(1);"ESPAÇAMENTO NA DIREÇÃO Y= ";EY;"(M)"
2310 LPRINT;TAB(1);"RELAÇÃO CORTE/ATERRO (R)=";R
2320 END

```

O SOLO

ÓRGÃO OFICIAL DO CENTRO ACADÊMICO "LUIZ DE QUEIROZ"



ESALQ/USP

ISSN 0584 - 08211 . PIRACICABA - SP . VOL. 80 . Nº 1 E 2 . P.1-82 . JAN/DEZ . 1995